

```
{==========================================================================================}
{==================================  参数  ==============================================}
{==========================================================================================}
P1 := 5;
P2 := 10;
P3 := 60;
P4 := 120;
N  := 27;
NFS := 11;

{SHOWBKG  := 1;}
{==========================================================================================}
{==========================================================================================}
{==================================  定义  ==============================================}
{==========================================================================================}
{==========================================================================================}
{MA}
MA1 := EMA(CLOSE,P1),DRAWNULL;
MA2 := EMA(CLOSE,P2),DRAWNULL;
MA3 := EMA(CLOSE,P3),DRAWNULL;
MA4 := EMA(CLOSE,P4),DRAWNULL;
CUPMA1 := MA1>=REF(MA1,1);
CUPMA2 := MA2>=REF(MA2,1);
CUPMA3 := MA3>=REF(MA3,1);
CUPMA4 := MA4>=REF(MA4,1);

{STD}
SD     := EMA(STD(CLOSE,N), 2);
{SD    := EMA(SD,3);}
CUPStd := SD >= REF(SD,1);
CDNStd := SD <= REF(SD,1) ;

{MID}
MID := MA(CLOSE,N);
CUPMID := MID>=REF(MID,1);


{BOLL}
BUUU := MID + 3*SD;
BUU  := MID + 2*SD;
```

```
BU   := MID + 1*SD;
BL   := MID - 1*SD;
BLL  := MID - 2*SD;
BLLL := MID - 3*SD;
CUPBUUU := BUUU >= REF(BUUU,1);
CDNBLLL := BLLL <= REF(BLLL,1);


{MACD}
DIF := 100* (EMA(C,12)/EMA(C,26)-1);
DEA := EMA(DIF,9);
MACD := DIF-DEA;
MMACD := EMA(MACD,2);
CUPDIF := DIF>=REF(DIF,1);
CUPDEA := DEA>=REF(DEA,1);
CUPMACD := MACD>=REF(MACD,1);
CUPMMACD := MMACD>=REF(MMACD,1);


{KDJ}
RSV = (CLOSE-LLV(LOW,9))/(HHV(HIGH,9)-LLV(LOW,9))*100;
K := SMA(RSV,3,1);
D := SMA(K,3,1);
J := 3*K-2*D;
CUPK := K>=REF(K,1);
CUPD := D>=REF(D,1);
CUPJ := J>=REF(J,1);


CPBX := EMA(C,2);
CPSX := EMA(SLOPE(C,21)*20 + C,42);
CUPCPSX := CPSX>=REF(CPSX,1);


{BBI}
BBI:= EMA((EMA(CLOSE,3)+EMA(CLOSE,6)+EMA(CLOSE,12)+EMA(CLOSE,24))/4,2);
CUPBBI := BBI>=REF(BBI,1);

{UU}
UU := (C>MA1 AND C>MA2) AND C>MID AND C>MA3, DRAWNULL, COLORGRAY;

{-------------------------------- K self -----------------------------------------}
IH := MAX(OPEN, CLOSE);
IL := MIN(OPEN, CLOSE);
IDD := IH - IL;
ITT := H - L;
IM  := (H+L+O+C)/4;

IPMove := 100 * (c/ref(c,1)-1);
IPAbs := 100 * IDD/ref(c,1);
IPOsc := 100 * ITT/ref(c,1);

{-------------------------------- MA层 -------------------------------------------}
LMAreal := (MA1 - MID)/(1.9*SD);
```

```
LMA := MIN(1, MAX(0, LMAreal));
CUPLMA := LMA>=REF(LMA,1);
CDNLMA := LMA<REF(LMA,1);

{--------------------------------  C层  -------------------------------------------}
LCreal := (C-MID)/(1.98*SD);
LC := MAX(0, LCreal);


{========================================================================================}
{==================================  特殊信号  ========================================}
{========================================================================================}
{--------------------------------  入轨  ------------------------------------------}
{实体大于BUU部分/实体}
TPC2DD := IF(
  IH > IL,
  ( MAX(IH, BUU) - MAX(IL, BUU) ) / IDD,
  IF(IH>=BUU AND o>ref(c,1),  1,  0)
);

{实体大于BUU部分/通道}
TPC2SD := ( MAX(IH, BUU) - BUU ) / SD;
{最高大于BUU部分/通道}
TPH2SD := ( MAX(H, BUU) - BUU ) / SD;

TTNC :=  COUNT(TPC2SD>0, 5);
TTNH :=  COUNT(TPH2SD>0, 5);

TTIS :=
TPC2DD>0
OR (C>MA1 AND TTNH>2)
OR (C>MA1 AND TTNC>1);

{========================================================================================}
{========================================================================================}
{==================================  信号系统  ========================================}
{========================================================================================}
{========================================================================================}
{--------------------------------  操盘线系统  ------------------------------------}
VCPX := CPX>0 and C>MID;
{--------------------------------  指导线系统  ------------------------------------}
VBBI := MA1>BBI;
{--------------------------------  ZACD系统  --------------------------------------}
{最初试探}
VZACD0:=
 DIF<=0
 AND MMACD>0
 {AND CUPMMACD}
 AND MA1>BBI
 AND ALL(C>MA1,BARSLAST(CROSS(MACD,-0.1)))
 AND C>MA2
,DRAWNULL;
```

```
{最正常上升}
VZACDMainup:=
 DIF>-1
 AND (
  ALL(CUPDIF,BARSLAST(CROSS(MACD,-0.1)))
  OR CROSS(DIF,0)
  OR CROSS(DIF,DEA)
)
,DRAWNULL;

{二次拉升}
VZACDBeforeJC:=
 DEA>0
 AND count(CUPDIF=0, BARSLAST(CROSS(DIF,0)))>1
 AND (UU OR REF(UU,1)=1)
 AND BETWEEN(MACD,-2,0.1)
 AND CUPMMACD
 AND CUPDIF
,DRAWNULL;

VZACD2up:=
 DEA>0
 AND count(CUPDIF=0, BARSLAST(CROSS(DIF,0)))>1
 AND (UU OR REF(UU,1)=1)
 AND CUPDIF
 AND MACD>0
,DRAWNULL;


{==============================  仪轨 ==========================================}
{持 : VBBI
AND CUPBUUU AND (CUPStd OR REF(CUPStd,1)=1)
AND LMA>0.5
AND (LC>0.5 AND C>BBI AND COUNT(LC>1, 10)>0)
,DRAWNULL, COLORWHITE;}

持 : VBBI
AND CUPBUUU
AND LMA>0
AND (VZACDMainup OR (LMA>0.51 AND (LC>0.5 AND C>BBI AND COUNT(LC>1, 10)>0)))
,DRAWNULL, COLORWHITE;

观 : VBBI
AND (CDNStd OR (CUPStd AND COUNT(CUPStd, 10)<7))
AND LMA>0.1
AND (VZACDBeforeJC OR VZACDMainup OR VZACD2up)
,DRAWNULL, COLORWHITE;

安 : VCPX
,DRAWNULL, COLORWHITE;

{==============================================================================}
{==============================================================================}
{==============================  Make Up  =======================================}
```

```
{==================================================================================}
{==================================================================================}
{-------------------------------- 区域 ------------------------------------------}
{
全部提示区域：RGNx
操作区域：RGNo
方差区域：RGNstd
MABU区域：RGNm
特信区域：RGNs
验证区域：RGNt}

RGNxU := MID;
RGNxL := BL;
II := RGNxU - RGNxL;

RGNoU := RGNxU - 0*II/10;
RGNoL := RGNxU - 2*II/10;
RGNmU := RGNxU - 3*II/10;
RGNmL := RGNxU - 7*II/10;

RGNSchgU := RGNxU - 7*II/10;
RGNSchgL := RGNxU - 8*II/10;

RGNstdU := RGNxU - 8*II/10;
RGNstdL := RGNxL;

{==================================================================================}
{=============================== 背景 ==========================================}
{==================================================================================}
{-------------------------------- 全背景 -----------------------------------------}
DRAWGBK(1, RGB(0, 0, 0))LAYER7;

{-------------------------------- 仪轨 ------------------------------------------}
DRAWGBK(SHOWBKG=1 AND 安, RGB(0, 15, 30));
DRAWGBK(SHOWBKG=1 AND 观, RGB(0, 30, 50));
DRAWGBK(SHOWBKG=1 AND 持, RGB(0, 50, 80));

{==================================================================================}
{=============================== 周期 ==========================================}
{==================================================================================}
{收盘时刻}
VERTLINE(SHOWTIME=1 and (type=1 or STKLABEL='000300') and datatype<7 and TIME>145900,0)colorblack;
VERTLINE(SHOWTIME=1 and STRLEFT(STKLABEL,1)='I' and datatype<8 and TIME>151400,0)colorblack;
VERTLINE(SHOWTIME=1 and STRLEFT(STKLABEL,2)='HK' and datatype<8 and TIME>155900,0)colorblack;

{周开始}
VERTLINE(SHOWTIME=1 and type=1 and datatype<=8 and thisweek=1,0),color00090F;

{月开始}
VERTLINE(SHOWTIME=1 and type=1 and datatype<=9 and thismonth=1,0),color00090F;

{年开始}
VERTLINE(SHOWTIME=1 and type=1 and datatype<=10 and thisyear=1,0),color00090F;
```

```
{停牌}
TP:= DATETOD1970(DATE) - DATETOD1970(REF(DATE,1)) > 7;
{VERTLINE(type=1 and datatype<=9 and TP,0),colorGRAY;}
stickline(type=1 and datatype<=9 and TP, RGNxL*0.99, RGNxL, 0.5,0),colorGRAY;

{====================================================================================}
{================================ 地上 ===========================================}
{====================================================================================}
{-------------------------------- 入轨信号 ---------------------------------------}
{
FILLRGN(BUU,BUUU,
TTIS, RGB(20, 50, 90),
(FF>=BUUU OR H>BUU) AND CUPFF,RGB(0, 50, 90)  );
FILLRGN(BUUU-SD/10, BUUU, TTIS, RGB(255, 50, 0));}
DRAWTEXT(TPC2DD=1 and TPC2SD<1, MID, '*'),ALIGN1,VALIGN2,Colorgreen;


{信 地产以及各种启动模式，确定是否进入轨道}
{VERTLINE(C>=BUU AND BETWEEN(COUNT(C>=BUU,12),3,3),0),colorred; }

{====================================================================================}
{================================ 地下 ===========================================}
{====================================================================================}
{-------------------------------- ZACD -------------------------------------------}
{FILLRGN(RGNoL, RGNoU,
 VZACDMainup, RGB(50, 139, 50),
 VZACD2up, RGB(120, 50, 50),
 VZACDBeforeJC, RGB(200, 180, 0),
 VZACD0, RGB(255, 150, 0)
);}

{FILLRGN(RGNoL, RGNoU,   CPX>0, RGB(50, 160, 50) );}



{-------------------------------- Std --------------------------------------------}
{FILLRGN(RGNstdL, RGNstdU,
CUPStd AND C>MID, RGB(0, 70, 50),
1, RGB(0, 50, 50));}

PARTLINE(BL,
CUPStd AND C>MID, RGB(0, 70, 50),
1, RGB(0, 50, 50)
)linethick3;
{====================================================================================}
{================================ 布线 ===========================================}
{====================================================================================}
PARTLINE(BUUU, 1,RGB(0, 20, 60));
PARTLINE(BUU, 1,RGB(0, 20, 60));

PARTLINE(MID, 1, RGB(55, 55, 55));

PARTLINE(MID,
CPX>0,RGB(50, 160, 50),
```

```
CPX=-1,RGB(200, 0, 0)
)linethick3;

PARTLINE(MA1, 1, RGB(255, 255, 0));
PARTLINE(BBI , 1,RGB(0, 150, 250));


{================================= 提示 ==============================================}
{DRAWTEXT(SHOWBS=1 AND cross(CPBX,CPSX),LOW,'cB'),ALIGN1,VALIGN0,colorred;
DRAWTEXT(SHOWBS=1 AND cross(CPSX,CPBX),HIGH,'cS'),ALIGN1,VALIGN2,colorgreen; }
{========================================================================================}
{================================= K线 ===============================================}
{========================================================================================}

{BBI-MA1}
STICKLINE(C>BBI AND MA1>BBI,  BBI,MA1,    4,0),COLOR443300;
{MA1-BUU}
STICKLINE(C>MA1 AND MA1>BBI,   MA1,C,   4,0),COLOR665500;
{BUU-BUUU}
stickline(C>BUU, BUU,C,  4,0),COLORFFAA00;
{BUUU-}
stickline(C>BUUU,  BUUU,C,  4,0)colorEE00EE;

{反攻}
{
STICKLINE(C>ZJM AND ZJ<ZK,  ZJM,C,    3,0)COLORGRAY;
STICKLINE(C>ZJM AND ZJ<ZK AND ZJ>ZJM and CUPZJJM,  O,C,    7,0)COLORMAGENTA;}


{跑得快}
STICKLINE(C<MA1 AND MA1>BBI,   MA1,MIN(MIN(O,C),MA1),   6,0)COLORRED;


{========================================================================================}
{================================= 赋值 ==============================================}
{========================================================================================}

C,DRAWNULL, COLORCYAN;

突破 : TTIS, DRAWNULL, COLORyellow;
突数 : TTNC, DRAWNULL, COLORyellow;
突自 : 100*TPC2DD, DRAWNULL, COLORyellow;
突天 : 100*TPC2SD, DRAWNULL, COLORyellow;

五线 : if(CPBX>=MA1, MA1, 0), DRAWNULL, COLORWHITE;
指导线:BBI, DRAWNULL, COLORgray;
U : UU , DRAWNULL, COLORWHITE;
均层 : INTPART(100*LMA), DRAWNULL, COLORwhite;
振幅 : 100*IPOsc, DRAWNULL,  DRAWNULL, COLORwhite;

IFS(CUPStd>0, '扩张','收缩'), DRAWNULL, COLORblue;
IFS(CUPBUUU>0, '轨升','轨降'), DRAWNULL, COLORblue;
IFS(C>MA3, '60日均线之上','小于60日均线'), DRAWNULL, COLORwhite;
IFS(C>MA4, '120日均线之上','小于120日均线'), DRAWNULL, COLORwhite;
```

```
{--------------------------------- 鸟瞰 -------------------------------------------}
STICKLINE(SHOWBIRD>1, HHVALL(H)*SHOWBIRD, HHVALL(H)*SHOWBIRD, 1,0)colorblack;

{==============================================================================}
{==============================================================================}
{==============================================================================}
```