```
{=====================================================================================}
{================================ 参数 ============================================}
{=====================================================================================}
{----------| 颜色 ------------------------------------------------------------------
MA1    Yellow           RGB(250, 250, 0)    ColorFAFA00
MA1<    Darkyellow    RGB(100, 100, 0)    ColorC8C800
MA2    Red          RGB(255, 50, 0)        ColorFF3200
MA2<    DarkRed        RGB(100, 0, 0)           Color640000
MID    White           RGB(125,125,125)     Color7D7D7D
DKXB    LightGreen    RGB(144, 238, 144)    Color90EE90
BU    DeepPink    RGB(255, 20, 147)    ColorFF1493
BUU    DarkGoldenrod1    RGB(255, 185, 15)    ColorFFB90F
-------------------------------------------------------------------------------------}
P1 := 5;
P2 := 10;
P3 := 60;
P4 := 120;
N  := 26;

NDKSD := 10;
NDKLD := 20;
Ndkdea:= 6;

C,drawnull;
{=====================================================================================}
{================================ 定义 ============================================}
{=====================================================================================}
{MA}
MA1 := MA(CLOSE,P1),DRAWNULL;
MA2 := MA(CLOSE,P2),colorred;
MID := MA(CLOSE,N),DRAWNULL;
MA3 := MA(CLOSE,P3),color68228B;
MA4 := MA(CLOSE,P4),colorcyan;
CUPMA1:=MA1>=REF(MA1,1);
CUPMA2:=MA2>=REF(MA2,1);
CUPMA3:=MA3>=REF(MA3,1);
CUPMA4:=MA4>=REF(MA4,1);
CUPMID:=MID>=REF(MID,1);
CDNMA1:=MA1<=REF(MA1,1);
CDNMA2:=MA2<=REF(MA2,1);
CDNMA3:=MA3<=REF(MA3,1);
CDNMA4:=MA4<=REF(MA4,1);
CDNMID:=MID<=REF(MID,1);

{DK}
A := (3*C+L+O+H)/6;
DKXB := (20*A+19*REF(A,1)+18*REF(A,2)+17*REF(A,3)+16*REF(A,4)+15*REF(A,5)+14*REF(A,6)
+13*REF(A,7)+12*REF(A,8)+11*REF(A,9)+10*REF(A,10)+9*REF(A,11)+8*REF(A,12)
+7*REF(A,13)+6*REF(A,14)+5*REF(A,15)+4*REF(A,16)+3*REF(A,17)+2*REF(A,18)+
REF(A,20))/210;
DKSD := MA(DKXB,NDKSD);
DKLD := MA(DKXB,NDKLD);

CUPDKXB :=DKXB>=REF(DKXB,1);
CUPDKSD :=DKSD>=REF(DKSD,1);
```

```
CUPDKLD :=DKLD>=REF(DKLD,1);

{SHORT DKX-MACD}
DKdif    := MIN(100,100* (DKXB/DKSD-1)),COLORWHITE;
DKdea    := EMA(DKdif,Ndkdea);
DKmacd   := (DKdif-DKdea),drawnull;
DKmmacd  := SMA(DKmacd,4,2);
CUPDKdif := DKdif >= REF(DKdif,1);
CUPDKdea := DKdea >= REF(DKdea,1);
CUPDKmmacd := DKmmacd >= REF(DKmmacd,1);

{LONG DKX-MACD}
DKLdif    := MIN(200,100* (DKXB/DKLD-1));
DKLdea    := EMA(DKLdif,NDKdea);
DKLmacd   := (DKLdif-DKLdea),drawnull;
DKLmmacd  := SMA(DKLmacd,4,2);
CUPDKLdif := DKLdif >= REF(DKLdif,1);
CUPDKLdea := DKLdea >= REF(DKLdea,1);
CUPDKLmmacd := DKLmmacd >= REF(DKLmmacd,1);

{STD}
StdC     := STD(CLOSE,N);
StdMA    := MA(StdC,2);
CUPStdMA := StdMA >= REF(StdMA,1)*0.98;
CDNStdMA := StdMA <= REF(StdMA,1) ;

{BOLL}
BUU := MID + 2*StdC;
BU  := MID + 1*StdC;
BL  := MID - 1.5*StdC;
BLL := MID - 2*StdC;
CUPBUU :=BUU>=REF(BUU,1);
CUPBU  :=BU>=REF(BU,1);

{MACD}
DIF := 100* (EMA(C,12)/EMA(C,26)-1);
DEA := EMA(DIF,9);
MACD := (DIF-DEA);
MMACD := SMA(MACD,4,2);
CUPDIF := DIF>=REF(DIF,1);
CUPDEA := DEA>=REF(DEA,1);
CUPMACD := MACD>=REF(MACD,1);
CUPMMACD := MMACD>=REF(MMACD,1);

{KDJ}
RSV = (CLOSE-LLV(LOW,9))/(HHV(HIGH,9)-LLV(LOW,9))*100;
K := SMA(RSV,3,1);
D := SMA(K,3,1);
J := 3*K-2*D;
CUPK := K>=REF(K,1);
CUPD := D>=REF(D,1);
CUPJ := J>=REF(J,1);

{VOLUME}
VMA1:=MA(VOL,5);
VMA2:=MA(VOL,10);
```

```
VMA3:=MA(VOL,20);
CUPVMA1:=VMA1>REF(VMA1,1);
CUPVMA2:=VMA2>REF(VMA2,1);
CUPVMA3:=VMA3>REF(VMA3,1);

{CCI}
TYP := (HIGH + LOW + CLOSE)/3;
CCI := (TYP-MA(TYP,26))/(0.015*AVEDEV(TYP,26));

{SYS}
JJJ:=IF(DYNAINFO(8)>0.01,0.01*DYNAINFO(10)/DYNAINFO(8),DYNAINFO(3));
DDD:=(DYNAINFO(5)<0.01 || DYNAINFO(6)<0.01);
JJJT:=IF(DDD,1,(JJJ<(DYNAINFO(5)+0.01) && JJJ>(DYNAINFO(6)-0.01)));
CYC1:=IF(JJJT,0.01*EMA(AMOUNT,P1)/EMA(VOL,P1),EMA((HIGH+LOW+CLOSE)/3,P1));
CYC∞:=IF(JJJT,DMA(AMOUNT/(100*VOL),100*VOL/FINANCE(7)),EMA((HIGH+LOW+CLOSE)/3,120));

{================================ 基础线 ==========================================}
PARTLINE(MA3,1,RGB(160, 32, 240));
PARTLINE(MA4,1,RGB(0, 255, 255));
PARTLINE(DKXB,1,RGB(0, 255, 0));
FILLRGN(BLL,BUU,CUPMID ,RGB(40, 40, 40));
{==================================================================================}
{================================ 周期分割 ========================================}
{==================================================================================}
VERTLINE(G=1 and (type=1  or STKLABEL='000300') and datatype<7 and TIME>145900,0),LINETHICK1;
VERTLINE(G=1 and (STRLEFT(STKLABEL,1)='I' ) and datatype<8 and TIME>151400,0),LINETHICK1;
VERTLINE(G=1 and STRLEFT(STKLABEL,2)='HK' and datatype<8 and TIME>155900,0),LINETHICK1;

VERTLINE(0 and type=1 and datatype=8 and weekday=5,0),LINETHICK2;

CTM := ((type=1 AND TIME>144500)  AND BETWEEN(DATATYPE,5,7) )
OR (STRLEFT(STKLABEL,2)='HK' AND BETWEEN(DATATYPE,5,7) and TIME>155900);


{==================================================================================}
{================================ TT ==============================================}
{==================================================================================}
{----------|  阶段0+|层-2<BLL,BL>：突破所有均线TT  -----------------------------------}
{----------|  MA1|MA2 DKXB|MA3 MA4     :     XXBXX  -----------------------------------}

{----------|  TT:大于DKXB ------------------------------------------------------------}
TTXXBXX := C>DKXB AND C>DKSD AND C>MID,DRAWNULL;
FILLRGN(BLL,BL-5*(BL-BLL)/6,TTXXBXX ,RGB(0, 191, 255));



{----------|  TT:大于MA3 -------------------------------------------------------------}
TTXXB1X:= TTXXBXX AND C>MA3,DRAWNULL;
TT0XB1X:= TTXXB1X AND C<MA1,DRAWNULL;
TTX0B1X:= TTXXB1X AND C<MA2,DRAWNULL;
FILLRGN(BL-5*(BL-BLL)/6,BL-1*(BL-BLL)/2,TTXXB1X,RGB(0, 154, 205));
FILLRGN(BL-5*(BL-BLL)/6,BL-1*(BL-BLL)/2,TT0XB1X,RGB(100, 100, 0));
FILLRGN(BL-5*(BL-BLL)/6,BL-1*(BL-BLL)/2,TTX0B1X,RGB(100, 0, 0));

{----------|  TT:大于MA4 -------------------------------------------------------------}
TTXXB11 := TTXXB1X AND C>MA4,DRAWNULL;
TT0XB11 := TTXXB11 AND C<MA1,DRAWNULL;
```

```
TTX0B11 := TTXXB11 AND C<MA2,DRAWNULL;
FILLRGN(BL-1*(BL-BLL)/2,BL,TTXXB11,RGB(0, 104, 139));
FILLRGN(BL-1*(BL-BLL)/2,BL,TT0XB11,RGB(100, 100, 0));
FILLRGN(BL-1*(BL-BLL)/2,BL,TTX0B11,RGB(100, 0, 0));

{=================================================================================}
{================================ 阶段 整理 ========================================}
{=================================================================================}
{----------| 层-3<BLL以下>：Std变化 ------------------------------------------------}
FILLRGN(BLL-(BL-BLL)/20,BLL,
CDNStdMa, RGB(60, 60, 60),
CUPStdMa, RGB(238, 44, 44)
);


{----------| 阶段-1|层-3<BLL以下>：收敛阶段 ----------------------------------------}
WMWM:= CDNStdMa AND BETWEEN(C,BL,BUU)
AND (CUPDKmmacd or (DKmmacd>0 and CUPDKdif and CUPDKdea))
,DRAWNULL,COLORGRAY;
FILLRGN(BLL-(BL-BLL)/5,BLL, WMWM, RGB(139,10,80));

{----------| 阶段1|层-3<BLL以下>：收敛阶段反攻的层次 -------------------------------}
FILLRGN(BLL-1*(BL-BLL)/10,BLL-(BL-BLL)/20,
WMWM and C>BU,RGB(0, 64, 128),
WMWM and C>MID,RGB(125, 125, 125),
WMWM and C>DKXB,RGB(0, 250, 0),
WMWM and C>MA2,RGB(255, 50, 0),
WMWM and C>MA1,RGB(250, 250, 0)
);

OO0:= TTXXB11 AND WMWM AND CPX>0,DRAWNULL,COLORGRAY;

{=================================================================================}
{================================ 阶段 上升 ========================================}
{=================================================================================}
{----------| 阶段2+3|层-3<BLL以下>：C突破BOLL上轨 ----------------------------------}
OO2 := CUPStdMa AND C>BU AND COUNT(C>BUU,5)=0,DRAWNULL,COLORGRAY;          {阶段2}
OO3 := CUPStdMa AND C>BUU,DRAWNULL,COLORGRAY;                              {阶段3}
FILLRGN(BLL,BL,
OO2, RGB(205, 102, 0),
OO3, RGB(255, 215, 0)
);

{----------| 阶段4|层1<MID,DKXB>：主升浪，MA1>=BU ----------------------------------}
OO4 := MA1>=BU ,DRAWNULL,COLORGRAY;                                        {阶段4}
FILLRGN(MID,DKXB,
OO4 AND COUNT(C<MA2,2)>1,RGB(139, 129, 76),
OO4,RGB(255, 215, 0)
);
FILLRGN(MID,MID+2*(DKXB-MID)/3,MA1>=BU AND C<MA1,RGB(100, 100, 0));
FILLRGN(MID,MID+2*(DKXB-MID)/3,MA1>=BU AND C<MA2,RGB(100, 0, 0));


{=================================================================================}
{================================ 特殊信号 ========================================}
{=================================================================================}
```

```
wDKJinchaqian:= C>MID AND DKdif>-2 AND BETWEEN(DKmacd,-0.9,0.1) ,drawnull;
wDKJincha:= C>MID AND DKdif>-2 AND CROSS(DKdif,DKdea),drawnull;



{==========================================================================================}
{==========================================================================================}
{==================================== 信号系统 ============================================}
{==========================================================================================}
{==========================================================================================}
{----------| 均线系统 -------------------------------------------------------------}
BMA :=
DKXB>DKSD
AND MA2>MID
AND MA2>MA3
,DRAWNULL,COLORGRAY;
FILLRGN(BL,BL+1.2*(MID-BL)/9,BMA,RGB(85, 26, 139));

{----------| ZBDK系统 -------------------------------------------------------------}
BZBDK :=
CUPDKXB
AND (CUPDKDIF OR (DKDIF>0 AND BETWEEN(DKMACD,-0.3,0.5)))
,DRAWNULL,COLORGRAY;
FILLRGN(BL+1.2*(MID-BL)/9,BL+3*(MID-BL)/9,BZBDK,RGB(155, 48, 255));

{----------| 操盘线系统 -------------------------------------------------------------}
BCPX := CPX>0,,DRAWNULL,COLORGRAY;
FILLRGN(BL+3*(MID-BL)/9,BL+6*(MID-BL)/9,BCPX,RGB(152, 251, 152));

{----------| BS系统 -------------------------------------------------------------}
{----------| 交易量系统 -------------------------------------------------------------}


{----------| Shark系统 -------------------------------------------------------------}
BMY :=
MA1>MID AND
C>MA2 AND C>MID AND CUPMA1 AND CUPMA2
AND DKdif>-5
AND DKmacd>-1
AND (
CUPDKMMACD
OR CUPDKdif
OR (DKdea>3 AND CUPMMACD)
),DRAWNULL,COLORGRAY;
FILLRGN(MID-1*(MID-BL)/3,MID,BMY,RGB(3, 110, 184));

STICKLINE(BMY AND BETWEEN(COUNT(C>BU,5),1,3) AND C<BUU,H,H+(H-L)/8,7,0)colorYELLOW;

{==========================================================================================}
{==================================== 选股 ============================================}
{==========================================================================================}
买:
(BMY OR (BCPX AND BMA))
AND BZBDK
AND (OO2 OR OO3 OR OO4)
,DRAWNULL,COLORYELLOW;
```

```
抢:
BMY AND BCPX
AND OO4
,DRAWNULL,COLORYELLOW;


看:
(BZBDK AND BMA)
OR (OO2 OR OO3 OR OO4)
OR BMY
OR OO0
,DRAWNULL,COLORYELLOW;


丢:COUNT(C<MA1,2)=2,DRAWNULL,COLORRED;


{=========================================================================================}
{=================================  布线  ==================================================}
{=========================================================================================}
{----------|  均线  ----------------------------------------------------------------------}

PARTLINE(BU, 1, RGB(0, 64, 128));
PARTLINE(BUU, 1, RGB(0, 64, 128));
PARTLINE(MID, 1, RGB(125, 125, 125))LINETHICK2;
PARTLINE(MA1,
CUPMA1,RGB(250, 250, 0),
1, RGB(200, 200, 0)
);
PARTLINE(MA2,
CUPMA2,RGB(255, 50, 0),
1, RGB(100, 0, 0)
);



{----------|  赋值  ----------------------------------------------------------------------}
买自 : BMY, DRAWNULL, COLORWHITE;
买操 : BCPX, DRAWNULL, COLORWHITE;
买周 : BZBDK, DRAWNULL, COLORWHITE;
买均 : BMA, DRAWNULL, COLORWHITE;
破均 : OO4, DRAWNULL, COLORYELLOW;
破布二 : OO3, DRAWNULL, COLORYELLOW;
破布一 : OO2, DRAWNULL, COLORYELLOW;
收敛 : OO0, DRAWNULL, COLORGRAY;
UU : TTXXBXX AND (C>MA1 AND C>MA2) AND C>MA3, DRAWNULL, COLORGRAY;
M1 : MA1, DRAWNULL, COLORGRAY;
M2 : MA2, DRAWNULL, COLORGRAY;

{----------|  鸟瞰  ----------------------------------------------------------------------}
STICKLINE(BIRDVIEW AND UU,C*BIRDVIEW ,C*BIRDVIEW ,9,0)colorYELLOW;
{=========================================================================================}
{=========================================================================================}
{=========================================================================================}
```