```
{==================================================================================}
{=============================== 参数 ==========================================}
{==================================================================================}
P1 := 5;
P2 := 10;
P3 := 60;
P4 := 120;
N  := 27;

NDKSD := 10;
NDKLD := 20;
Ndkdea:= 6;

SHOWBKG  := 1;
SHOWTT   := 0;

{C,drawnull;}
{==================================================================================}
{================================ 定义 =========================================}
{==================================================================================}
{MA}
MA1 := MA(CLOSE,P1),DRAWNULL;
MA2 := MA(CLOSE,P2),DRAWNULL;
MID := MA(CLOSE,N),DRAWNULL;
MA3 := MA(CLOSE,P3),DRAWNULL;
MA4 := MA(CLOSE,P4),DRAWNULL;
CUPMA1:=MA1>=REF(MA1,1);
CUPMA2:=MA2>=REF(MA2,1);
CUPMA3:=MA3>=REF(MA3,1);
CUPMA4:=MA4>=REF(MA4,1);
CUPMID:=MID>=REF(MID,1);
CDNMA1:=MA1<=REF(MA1,1);
CDNMA2:=MA2<=REF(MA2,1);
CDNMA3:=MA3<=REF(MA3,1);
CDNMA4:=MA4<=REF(MA4,1);
CDNMID:=MID<=REF(MID,1);

{CPX}
CPBX := EMA(C,2);
CPSX := EMA(SLOPE(C,21)*20+C,42);
CPZDX:= EMA((EMA(CLOSE,4)+EMA(CLOSE,6)+EMA(CLOSE,12)+EMA(CLOSE,24))/4,2);
OOCP := MA1>CPZDX AND CPBX>=CPSX,DRAWNULL,COLORGRAY;

{DK}
A := (3*C+L+O+H)/6;
DKXB := (20*A+19*REF(A,1)+18*REF(A,2)+17*REF(A,3)+16*REF(A,4)+15*REF(A,5)+14*REF(A,6)
+13*REF(A,7)+12*REF(A,8)+11*REF(A,9)+10*REF(A,10)+9*REF(A,11)+8*REF(A,12)
+7*REF(A,13)+6*REF(A,14)+5*REF(A,15)+4*REF(A,16)+3*REF(A,17)+2*REF(A,18)+
REF(A,20))/210;
DKSD := MA(DKXB,NDKSD);
DKLD := MA(DKXB,NDKLD);

CUPDKXB :=DKXB>=REF(DKXB,1);
CUPDKSD :=DKSD>=REF(DKSD,1);
```

```
CUPDKLD :=DKLD>=REF(DKLD,1);

{SHORT DKX-MACD}
DKdif    := MIN(100,100* (DKXB/DKSD-1)),COLORWHITE;
DKdea    := EMA(DKdif,Ndkdea);
DKmacd   := (DKdif-DKdea),drawnull;
DKmmacd  := SMA(DKmacd,4,2);
CUPDKdif := DKdif >= REF(DKdif,1);
CUPDKdea := DKdea >= REF(DKdea,1);
CUPDKmmacd := DKmmacd >= REF(DKmmacd,1);

{LONG DKX-MACD}
DKLdif    := MIN(200,100* (DKXB/DKLD-1));
DKLdea    := EMA(DKLdif,NDKdea);
DKLmacd   := (DKLdif-DKLdea),drawnull;
DKLmmacd  := SMA(DKLmacd,4,2);
CUPDKLdif := DKLdif >= REF(DKLdif,1);
CUPDKLdea := DKLdea >= REF(DKLdea,1);
CUPDKLmmacd := DKLmmacd >= REF(DKLmmacd,1);

{STD}
StdC     := STD(CLOSE,N);
StdMA    := MA(StdC,2);
CUPStdMA := StdMA >= REF(StdMA,1);
CDNStdMA := StdMA <= REF(StdMA,1) ;

{BOLL}
BUUU := MID + 3*StdC;
BUU  := MID + 2*StdC;
BU   := MID + 1*StdC;
BL   := MID - 1*StdC;
BLL  := MID - 2*StdC;
BLLL := MID - 2.9*StdC;
CUPBUUU := BUUU >= REF(BUUU,1),drawnull;
CDNBLLL := BLLL <= REF(BLLL,1),drawnull;


{MACD}
DIF := 100* (EMA(C,12)/EMA(C,26)-1);
DEA := EMA(DIF,9);
MACD := (DIF-DEA);
MMACD := SMA(MACD,4,2);
CUPDIF := DIF>=REF(DIF,1);
CUPDEA := DEA>=REF(DEA,1);
CUPMACD := MACD>=REF(MACD,1);
CUPMMACD := MMACD>=REF(MMACD,1);

{KDJ}
RSV = (CLOSE-LLV(LOW,9))/(HHV(HIGH,9)-LLV(LOW,9))*100;
K := SMA(RSV,3,1);
D := SMA(K,3,1);
J := 3*K-2*D;
CUPK := K>=REF(K,1);
CUPD := D>=REF(D,1);
```

```
CUPJ := J>=REF(J,1);

{VOLUME}
VMA1:=MA(VOL,P1);
VMA2:=MA(VOL,P2);
VMA3:=MA(VOL,N);
CUPVMA1:=VMA1>REF(VMA1,1);
CUPVMA2:=VMA2>REF(VMA2,1);
CUPVMA3:=VMA3>REF(VMA3,1);

{CYC}
CB1 := EMA(if(finance(0),amount/volunit,V*(h+l+c)/3),P1)/EMA(VOL,P1);
CBB := DMA(C,if(V/CAPITAL/100<0.5,V/CAPITAL/100,0.5));
CUPCB1 := CB1>REF(CB1,1);

{================================ 基础线 ==========================================}
PARTLINE(MA3,1,RGB(60, 60, 60));
PARTLINE(MA4,1,RGB(40, 40, 40));


{=====================================================================================}
{================================ 区域划分 ========================================}
{=====================================================================================
操作区域：RGNo
方差区域：RGNd
主升区域：RGNc
突破区域：RGNa
MABU区域：RGNm
特信区域：RGNs
验证区域：RGNt}

RGNoU := MID;
RGNoL := BLL + 3*(MID-BLL)/8;
RGNmU := BLL + 3*(MID-BLL)/8;
RGNmL := BLL + 1*(MID-BLL)/8;
RGNdU := BLL + 1*(MID-BLL)/8;
RGNdL := BLL;

RGNcU := BLL + 5*(MID-BLL)/8;
RGNcL := BLL + 5*(MID-BLL)/8;


RGNaU := BLL;
RGNaL := BLL;

RGNsU := BU;
RGNsL := MID;

RGNtU := BUUU+StdC;
RGNtL := BUUU-StdC;


{=====================================================================================}
{================================ TT@RGNa ==========================================}
```

```
{================ MA1|MA2 DKXB|MA3 MA4 ： XXBXX ================================}
{==============================================================================}

{----------| TT:大于DKXB ----------------------------------------------------------}
TTXXBXX := C>DKXB AND C>DKSD AND C>MID,DRAWNULL;
FILLRGN(RGNaL, RGNaU-5*(RGNaU-RGNaL)/6, SHOWTT=1 AND TTXXBXX ,RGB(0, 191, 255));

{----------| TT:大于MA3 ----------------------------------------------------------}
TTXXB1X:= TTXXBXX AND C>MA3,DRAWNULL;
TT0XB1X:= TTXXB1X AND C<MA1,DRAWNULL;
TTX0B1X:= TTXXB1X AND C<MA2,DRAWNULL;
FILLRGN(RGNaU-5*(RGNaU-RGNaL)/6, RGNaU-3*(RGNaU-RGNaL)/6,SHOWTT=1 AND TTXXB1X,RGB(0, 154, 205));
FILLRGN(RGNaU-5*(RGNaU-RGNaL)/6, RGNaU-3*(RGNaU-RGNaL)/6,SHOWTT=1 AND TT0XB1X,RGB(100, 100, 0));
FILLRGN(RGNaU-5*(RGNaU-RGNaL)/6, RGNaU-3*(RGNaU-RGNaL)/6,SHOWTT=1 AND TTX0B1X,RGB(100, 0, 0));

{----------| TT:大于MA4 ----------------------------------------------------------}
TTXXB11 := TTXXB1X AND C>MA4,DRAWNULL;
TT0XB11 := TTXXB11 AND C<MA1,DRAWNULL;
TTX0B11 := TTXXB11 AND C<MA2,DRAWNULL;
FILLRGN(RGNaU-3*(RGNaU-RGNaL)/6, RGNaU,SHOWTT=1 AND TTXXB11,RGB(0, 104, 139));
FILLRGN(RGNaU-3*(RGNaU-RGNaL)/6, RGNaU,SHOWTT=1 AND TT0XB11,RGB(100, 100, 0));
FILLRGN(RGNaU-3*(RGNaU-RGNaL)/6, RGNaU,SHOWTT=1 AND TTX0B11,RGB(100, 0, 0));

UU : TTXXBXX AND (C>MA1 AND C>MA2) AND C>MA3, DRAWNULL, COLORGRAY;


{==============================================================================}
{==============================================================================}
{================================ 阶段@RGNa&RGNd ==================================}
{==============================================================================}
O0 := 0.8;
O1 := 1.3;
O2 := 2;
O3 := 3;

{================================ MA阶段 ==========================================}
MA := MA1;
LM := (MA1-MID)/(StdC*0.98), DRAWNULL, colorblue;
{-------------------------------- MA阶段II ---------------------------------------}
LM2 := LM>O1;
{-------------------------------- MA阶段I ----------------------------------------}
LM1 := LM>O0;


{================================ C阶段 ===========================================}
LC := (C-MID)/(StdC*0.98), DRAWNULL, colorblue;
{-------------------------------- C阶段4：C突破BUUU ------------------------------}
LC4 := LC>=O3;
{-------------------------------- C阶段3：C介于(BUU,BUUU) ------------------------}
LC3 := LC>=O2;
{-------------------------------- C阶段2：C介于(BU,BUU) --------------------------}
LC2 := BETWEEN(LC, O1, O2);
{-------------------------------- C阶段1：C介于(MID,BU) --------------------------}
LC1 := BETWEEN(LC, O0, O1);
```

```
{-------------------------------- C阶段0：C介于(MID,BU) -------------------------}
LC0:= CDNStdMa
AND BETWEEN(C, MID, BUU)
AND MA1>=CPZDX
AND (CUPDKmmacd or (DKmmacd>0 and CUPDKdif and CUPDKdea));


{================================ 画图 ============================================}
{-------------------------------- 势@背景 ---------------------------------------}
{全背景}
DRAWGBK(SHOWBKG=1, RGB(0, 18, 0));

{持}
{DRAWGBK(SHOWBKG=1 AND LM>0.9, RGB(0, 39, 78));}

DRAWGBK(SHOWBKG=1
AND (OOCP AND C>CBB)
AND (CUPBUUU AND CUPStdMA )
AND ((LM>1 AND LC>1.5) OR (LC>1.9 AND LM>0.9))
, RGB(0, 49, 79));

{运行区域}
FILLRGN(MID, BUUU,1, RGB(10, 0, 0));
FILLRGN(BLL, MID, 1, RGB(0, 10, 0));

{CB FOREVER}
FILLRGN(CBB, CBB*0.98, 1, RGB(255, 228, 255));

{与 持 相同}
CHI:= (CUPBUUU AND OOCP) and LM2 AND (LC3 OR LC2);

{-------------------------------- 周期 ------------------------------------------}
VERTLINE(SHOWTIME=1 and (type=1 or STKLABEL='000300') and datatype<7 and TIME>145900,0);
VERTLINE(SHOWTIME=1 and STRLEFT(STKLABEL,1)='I' and datatype<8 and TIME>151400,0);
VERTLINE(SHOWTIME=1 and STRLEFT(STKLABEL,2)='HK' and datatype<8 and TIME>155900,0);

VERTLINE(SHOWTIME=1 and type=1 and datatype<=8 and thisweek=1,0),color00090F;
VERTLINE(SHOWTIME=1 and type=1 and datatype<=9 and thismonth=1,0),color00090F;
VERTLINE(SHOWTIME=1 and type=1 and datatype<=10 and thisyear=1,0),color00090F;
{-------------------------------- Std@RGNd --------------------------------------}
FILLRGN(RGNdL, RGNdU,
CUPStdMa AND C>MID, RGB(0, 139, 69),
1, RGB(0, 40, 40)
);

{STD减少时BUUU与MA的方向运动，收敛状态的总结}

{-------------------------------- LM --------------------------------------------}
FILLRGN(RGNmL, RGNmU,
BETWEEN(LM, 1.6, 3), RGB(255, 215, 0),
BETWEEN(LM, 1.5, 1.6), RGB(245, 210, 0),
BETWEEN(LM, 1.4, 1.5), RGB(235, 205, 0),
BETWEEN(LM, 1.3, 1.4), RGB(225, 200, 0),
BETWEEN(LM, 1.2, 1.3), RGB(215, 195, 0),
```

```
BETWEEN(LM, 1.1, 1.2), RGB(205, 190, 0),
BETWEEN(LM, 1.0, 1.1), RGB(195, 185, 0),
BETWEEN(LM, 0.95, 1.0), RGB(185, 180, 0)
);
FILLRGN(RGNmL, RGNmL+(RGNmU-RGNmL)*1/2,
BETWEEN(LM, 0.9, 1.0), RGB(185, 180, 0),
BETWEEN(LM, 0.8, 0.9), RGB(170, 160, 0)
);
FILLRGN(RGNmL, RGNmL+(RGNmU-RGNmL)*1/4,
BETWEEN(LM, 0.6, 0.8), RGB(140, 110, 0),
BETWEEN(LM, 0.3, 0.6), RGB(130, 100, 0),
BETWEEN(LM, 0, 0.3), RGB(100, 80, 0)
);

{与MACD以及ZBDK结合，结合变色}

{-------------------------------- LC -------------------------------------------}
FILLRGN(RGNcL, RGNcU,
LC3, RGB(255, 215, 0),
LC2, RGB(205, 50, 0),
LC0, RGB(111, 0, 111)
);

{考虑如何过滤震荡情况}
{=====================================================================================}
{================================ 特殊信号 ========================================}
{=====================================================================================}
wDKJinchaqian:= C>MID AND DKdif>-2 AND BETWEEN(DKmacd,-0.9,0.1) ,drawnull;
wDKJincha:= C>MID AND DKdif>-2 AND CROSS(DKdif,DKdea),drawnull;


{=====================================================================================}
{=====================================================================================}
{================================ 信号系统 ========================================}
{=====================================================================================}
{-------------------------------- MACD系统 ---------------------------------------}
{需要重写}
OOMACD :=
CUPDIF
AND (
 CUPMMACD
 OR CUPDKDif
 OR (DKDif>2 AND BETWEEN(DKMACD,-0.5,0.1)){并且没有背离}
),DRAWNULL,COLORGRAY;
{-------------------------------- ZBDK系统 ---------------------------------------}
OOZBDK :=
CUPDKXB
AND (
 CUPDKMMACD
 OR CUPDKDif
 OR (DKDif>2 AND BETWEEN(DKMACD,-0.5,0.1)){并且没有背离}
),DRAWNULL,COLORGRAY;

{-------------------------------- MY ---------------------------------------------}
```

```
BMY :=
C>MA2 AND C>MID
AND CUPMA2
{AND MA2>MA3}
AND OOZBDK
,DRAWNULL,COLORGRAY;

{--------------------------------- 操盘线系统 -------------------------------------}
BCPX := CPX>0,DRAWNULL,COLORGRAY;

{================================ 画图 ============================================}
FILLRGN(RGNoL+2*(RGNoU-RGNoL)/3, RGNoU, BMY, RGB(100,181,246));
FILLRGN(RGNoL+1*(RGNoU-RGNoL)/3, RGNoL+2*(RGNoU-RGNoL)/3, BCPX, RGB(21,101,192));
FILLRGN(RGNoL, RGNoL+1*(RGNoU-RGNoL)/3, OOCP, RGB(78, 238, 148));


{==================================================================================}
{==================================================================================}
{================================ 选股 ============================================}
{==================================================================================}
{==================================================================================}

持: OOCP
AND CUPBUUU
AND (LM2 OR LC3 OR (LC2 AND CUPStdMa)),
DRAWNULL,COLORYELLOW;

买: OOCP
AND (BMY AND BCPX)
AND (LC2 OR LC3 OR LM2 OR LC0)
,DRAWNULL,COLORYELLOW;

观:
(CUPBUUU AND OOCP and (LC2 OR LC0)) {已验证}
OR ((CUPBUUU AND OOCP) and (LC2 AND (LM2 OR CUPStdMa))) {已验证}
OR BMY {已验证}
OR (LC3 OR LM2)
,DRAWNULL,COLORYELLOW;


卖:COUNT(C<CB1,2)=2
,DRAWNULL,COLORRED;

{--------------------------------- 画图 -------------------------------------------}
{FILLRGN(RGNtL+1*(RGNtU-RGNtL)/2, RGNtU, 买, RGB(22, 0, 22));}
FILLRGN(RGNtL, RGNtL+1*(RGNtU-RGNtL)/2, 观, RGB(0, 22, 55));


{==================================================================================}
{================================ 布线 ============================================}
{==================================================================================}
{
PARTLINE(RGNbU, 1, RGB(20, 20, 20));
PARTLINE(RGNaU, 1, RGB(20, 20, 20));
```

```
PARTLINE(RGNcU, 1, RGB(20, 20, 20));
PARTLINE(RGNeU, 1, RGB(20, 20, 20));
}



PARTLINE(BUUU, 1, RGB(0, 64, 128));
PARTLINE(BUU, 1, RGB(0, 64, 128));
{PARTLINE(BU, 1, RGB(0, 64, 128));}


PARTLINE(MID, 1, RGB(66, 66, 66))LINETHICK2;
PARTLINE(MA1, CUPMA1, RGB(255, 255, 0), 1, RGB(160, 160, 0));
{PARTLINE(CB1, CUPCB1, RGB(255, 255, 0), 1, RGB(160, 160, 0));}
PARTLINE(CPZDX, 1, RGB(0, 255, 0))LINETHICK1;



{================================ 提示 ============================================}
DRAWTEXT(SHOWBS=1 AND cross(CPBX,CPSX),LOW,'+'),ALIGN1,VALIGN0,colorred,linethick3;
DRAWTEXT(SHOWBS=1 AND cross(CPSX,CPBX),HIGH,'='),ALIGN1,VALIGN2,colorgreen,linethick3;
DRAWTEXT(SHOWBS=1 AND cross(H,BUUU),H,'-'),ALIGN1,VALIGN2,Color666666;
DRAWTEXT(SHOWBS=1 AND cross(BUUU,H),H,'-'),ALIGN1,VALIGN2,Color666666;
DRAWTEXT(SHOWBS=1 AND cross(H,BUU),H,'-'),ALIGN1,VALIGN2,Color666666;
DRAWTEXT(SHOWBS=1 AND cross(C,BUU),H,'-'),ALIGN1,VALIGN2,Color666666;

DRAWTEXT(SHOWBS=1 AND cross(CPBX,CPSX),RGNmL,'+'),ALIGN1,VALIGN0,Color666666;
DRAWTEXT(SHOWBS=1 AND cross(CPSX,CPBX),RGNmL,'='),ALIGN1,VALIGN0,Color666666;

{================================ K线 ============================================}
stickline(C>BUUU and O>BUUU,MIN(O,C),MAX(O,C),10,0),coloryellow;
stickline(C>BUUU and BETWEEN(O,BUU,BUUU),O,BUUU,10,0),colorred;
stickline(C>BUUU and O<BUU,BUU,BUUU,10,0),colorred;
stickline(BETWEEN(C,BUU,BUUU) and O>BUUU,MIN(O,C),MAX(O,C),10,0),coloryellow;
stickline(BETWEEN(C,BUU,BUUU) and BETWEEN(O,BUU,BUUU),O,C,10,0),colorred;
stickline(BETWEEN(C,BUU,BUUU) and O<BUU,BUU,C,10,0),colorred;



{VERTLINE(C>=BUU AND BETWEEN(COUNT(C>=BUU,12),3,3),0),colorred; }

{-------------------------------- 赋值 -------------------------------------------}
买自 : BMY, DRAWNULL, COLORWHITE;
操盘 : BCPX, DRAWNULL, COLORWHITE;

'LM['+LM+']' , DRAWNULL, COLORYELLOW;
'LC['+LC+']' ,DRAWNULL, COLORYELLOW;
NCUU: COUNT(C>BUU,10),DRAWNULL, COLORYELLOW;

分析 : CPBX>=CPSX, DRAWNULL, COLORWHITE;
扩张 : CUPSTDMA, DRAWNULL, COLORGRAY;
轨升 : CUPBUUU, DRAWNULL, COLORGRAY;
轨降 : CDNBLLL, DRAWNULL, COLORGRAY;
```

```
振幅 : 100*(H-L)/REF(C,1), DRAWNULL, COLORCYAN;

CC : C, DRAWNULL, COLORCYAN;
指导 : CPZDX, DRAWNULL, COLORwhite;
成本 : CB1, DRAWNULL, COLORwhite;
{
B2 : BUU, DRAWNULL, COLORBLUE;
B1 : BU, DRAWNULL, COLORBLUE;
M1 : MA1, DRAWNULL, COLORGRAY;
M2 : MA2, DRAWNULL, COLORGRAY;
M3 : MA3, DRAWNULL, COLORGRAY;
M4 : MA4, DRAWNULL, COLORGRAY;
}
{-------------------------------- 鸟瞰 ------------------------------------------}
STICKLINE(SHOWBIRD AND (持 OR 买 OR 观), C*SHOWBIRD, C*SHOWBIRD,1,0)colorblack;

{==================================================================================}
{==================================================================================}
{==================================================================================}
```